

Minicalc per lo ZX81

Un tabellone elettronico anche per lo ZX81

di Enrico Ferreguti

Ebbene sì: la mania dei tabelloni elettronici ha colpito anche lo ZX. Con questo programma potrete provare l'ebbrezza delle tabelle di dati più complicate che vi vengono in mente, lo potrete usare per calcoli lunghi e ripetitivi e potrete sbizzarrirvi con le proiezioni, pensando, tra l'altro, che state usando un programma che ricalca i famosissimi Visicalc, Ozz, ecc. e che non vi è costato niente.

Per coloro che non hanno idea di cosa sia un tabellone elettronico cerchiamo di chiarire la situazione.

Quante volte vi siete trovati davanti a delle tabelle e quante volte avete constatato l'inadeguatezza di penna e calcolatrice? Con il programma Minicalc potremo usare il nostro ZX per eseguire queste totalizzazioni.

Facciamo un esempio: dobbiamo

calcolare il bilancio familiare per un arco di 5 mesi. Includiamo le principali voci: luce, gas, telefono, vitto e varie; le organizziamo in una tabella includendo delle altre voci: entrate, uscite, riporto, e tot. entrate (entrate + riporto) (vedi fig. 1).

Adesso è solo necessario riempire le caselle e legarle logicamente fra loro. Se però una volta compilata la tabella volessimo modificare una voce, con carta e penna si sarebbe costretti a ricominciare tutto da capo. Con ZX e Minicalc bastano pochi secondi per vedere la situazione aggiornata.

Uso del programma

Dopo aver caricato il programma lo si lancia con RUN, dopodiché si incontra la fase di carica-

	MAR	APR	MAG	GIU	LUG
ENTRATE					
TOT. ENTRATE					
LUCE					
GAS					
TEL.					
VITTO					
VARIE					
USCITE					
RIPORTO					

Figura 1.

mento dati. Il programma comincia a mostrarvi sul video la tabella vuota e in fondo la frase

CARICAMENTO DATI con l'attesa di input.

Il programma sta aspettando le

coordinate della casella da riempire (prima la colonna e poi la riga). L'input della casella può essere nu-

```

10 DIM U$(9,9,9)
20 LET X=1
30 GOSUB 1000
40 PRINT AT 20,3;"CARICAMENTO
DATI ";
50 INPUT F
60 INPUT Z
65 IF F>9 OR Z>9 THEN GOTO 140
70 PRINT F,Z
80 INPUT U$(F,Z)
90 LET X=F
100 GOTO 30
140 DIM W(9,9)
150 FOR F=1 TO 9
155 FOR Z=1 TO 9
160 IF CODE U$(F,Z)>27 AND CODE
U$(F,Z)<38 THEN LET W(F,Z)=VAL
U$(F,Z)
170 NEXT Z
180 NEXT F
200 CLS
210 PRINT "ADESSO FERMO" IL
PROGRAMMA"
220 PRINT "PER PERMETTERTI DI M
ODIFICARE"
230 PRINT "LE RELAZIONI ALLA RI
GA 8000"
240 PRINT
250 PRINT "POI PER RICOMINCIARE
BATTI GOTO 300"
260 STOP
300 GOSUB 1000
304 REM ROUTINE CA
305 GOSUB 8000
307 FAST
310 FOR F=1 TO 9
320 FOR Z=1 TO 9
325 IF W$(F,Z)="" THEN GOTO 340
330 IF CODE W$(F,Z)>27 AND CODE
U$(F,Z)<38 THEN LET W$(F,Z)=STR
$(U$(F,Z))
340 NEXT Z
350 NEXT F
360 SLOW
370 GOSUB 1000
380 PRINT AT 20,0;"COMANDI ? (D
X,SX,CA,CH,SE,ST,SA)"
390 INPUT H$
400 IF H$="DX" THEN GOTO 2000
405 IF H$="SA" THEN GOTO 3000
410 IF H$="SX" THEN GOTO 2100
415 IF H$="ST" THEN GOTO 4000
420 IF H$="CA" THEN GOTO 305
425 IF H$="CH" THEN GOTO 2500
427 IF H$="SE" THEN GOTO 3500
430 INPUT F
440 INPUT Z
441 IF F>X+2 OR F<X THEN PRINT
AT 20,0;"FUORI SCHERMO : RIPETI
"
442 IF F>X+2 OR F<X THEN GOTO 4
30
445 LET H$=""
447 LET K#=W$(F,Z)
450 FOR H=1 TO 9
452 LET H$(H)=CHR$(CODE K$(H)+
128)
455 NEXT H
457 PRINT AT Z+2,10*((F=4 OR F
=7 OR F=1)+2*(F=2 OR F=5 OR F=8)
+3*(F=3 OR F=6 OR F=9))-1)+2;H$
470 IF CODE U$(F,Z)>27 AND CODE
U$(F,Z)<38 THEN GOTO 500
480 INPUT U$(F,Z)
490 GOTO 520
500 INPUT W(F,Z)
510 LET W$(F,Z)=STR$(W(F,Z))
520 GOTO 370
1000 LET A$=""
1010 LET A$(8)=CHR$(X+156)
1020 LET A$(17)=CHR$(X+157)

```

```

1030 LET A$(27)=CHR$(X+158)
1035 CLS
1040 LET B$="■:-----
-----:
1050 LET C$=":
1055 PRINT A$
1060 FOR Z=1 TO 9
1070 PRINT B$
1080 PRINT CHR$(Z+156);C$
1085 NEXT Z
1090 PRINT B$
1095 FOR F=1 TO 9
1100 PRINT AT F+2,2;W$(X,F);AT F
+2,12;W$(X+1,F);AT F+2,22;W$(X+2
,F)
1110 NEXT F
1200 RETURN
1999 REM ROUTINE DX
2000 LET X=X+1
2010 IF X+2>9 THEN LET X=7
2020 GOTO 370
2099 REM ROUTINE SX
2100 LET X=X-1
2110 IF X<1 THEN LET X=1
2120 GOTO 370
2499 REM ROUTINE CH
2500 PRINT AT 20,0;"QUALE CASELL
A CAMBI ?
2510 INPUT F
2520 INPUT Z
2530 PRINT AT 20,0;"
2540 PRINT AT 20,0;"LA CASELLA "
:F:" "Z:" "E"" = ?"
2550 INPUT H$
2560 IF CODE H$>27 AND CODE H$<3
8 THEN GOTO 2600
2570 LET W$(F,Z)=H$
2580 GOTO 370
2600 LET W(F,Z)=VAL H$
2610 GOTO 370
2999 REM ROUTINE SA
3000 PRINT AT 20,0;"CHE NOME DAI
A QUESTO LAORDO ? "
3010 INPUT H$
3020 PRINT AT 20,0;"PREPARA IL R
EGISTRATORE
3030 PAUSE 3E2
3040 PRINT AT 20,0;"AVVIALO CON
<RECORD>
3050 PAUSE 150
3060 SAVE H$
3070 GOTO 370
3499 REM ROUTINE SE
3500 PRINT AT 20,0;"QUALE SETTOR
E ?
3510 INPUT N
3520 LET X=N+3-2
3530 GOTO 370
4000 REM ROUTINE ST
4005 PRINT AT 20,0;"QUALE SETTOR
E STAMPO ?
4010 INPUT N
4012 LET Q=N+3-2
4015 PRINT AT 20,0;"QUAL""E"" IL
NOME DEL LAORDO?"
4017 INPUT H$
4018 LPRINT H$
4020 FOR F=1 TO 9
4030 LPRINT ":",W$(Q,F);":",W$(Q
+1,F);":",W$(Q+2,F)
4040 NEXT F
4050 GOTO 370
7900 REM NELLE RIGHE SOTTOSTANTI
REM SONO MEMORIZZATE LE
7950 REM REL. FRA CASELLE
8000 LET W(2,3)=W(2,5)+W(2,6)
8010 LET W(2,4)=W(2,2)+W(2,3)
8020 LET W(3,3)=W(3,5)+W(2,3)
8030 LET W(3,4)=W(3,2)+W(3,3)
8040 RETURN
8050 RETURN

```


merico o alfanumerico; nel primo caso la casella sarà operativa, cioè sarà significativa ai fini del calcolo stesso, mentre nel secondo caso sarà dedicata solo al commento. Come in una qualsiasi tabella può essere omessa, a scapito però della chiarezza. Per uscire da questa fase bisognerà fornire un numero maggiore di 9.

Quando il calcolatore chiede le coordinate della casella, il programma si ferma per dare la possibilità di cambiare o riscrivere le relazioni in riga 8000. Vediamo subito di cosa si tratta: quando facciamo una tabella del tipo descritto prima i legami esistenti tra una casella e l'altra sono logici ed evidenti per noi ma non al calcolatore: dobbiamo quindi insegnarglieli. Per esempio della tabella di prima sappiamo che alla voce USCITE dovremo scrivere la somma di LUCE, GAS, TEL, VITTO e VARIE e questo è ciò che vogliamo "insegnare" al computer. Il tabellone, nel calcolatore, è memorizzato come una matrice bidimensionale di 9x9 elementi, quindi, immaginando che marzo sia il mese, batteremo nella riga 8000:

LET W(2,8)=W(2,4)+W(2,5)+
+W(2,6)+W(2,7)

che per il computer significa che la casella 2,8 (ricordiamoci che la prima colonna è occupata dal commento) è uguale alla somma delle caselle 2,4, 2,5, 2,6 e 2,7.

Con questo abbiamo già fatto un'importante osservazione: le caselle TOT. ENTRATE, USCITE e RIPORTO sono caselle dipendenti, quindi non si possono modificare (dato che, ogni volta che eseguiamo l'aggiornamento, il valore della nostra modifica viene sostituito dal valore dell'attribuzione in riga 8000) e devono apparire in riga 8000 o seguenti.

Tutte le altre caselle sono indipendenti e si possono modificare (anzi si devono affinché il programma abbia senso) e da loro dipendono le caselle le cui attribuzioni compaiono in riga 8000 (caselle dipendenti).

Fatto questo si riparte con GOTO 300.

Apparirà sullo schermo la griglia del tabellone, il contenuto delle caselle e in fondo il messaggio COMANDI? (DX,SX,CA,CH,SE,ST,SA) il cui significato verrà discusso in seguito.

Per poter modificare le caselle dovremo battere NEW/LINE. Appariranno due richieste di input con cui inseriremo le coordinate della casella da cambiare (prima la riga poi la colonna).

La casella che vogliamo trasformare sarà visualizzata in inverso e si potrà passare alla modifica.

Dopo aver battuto NEW/LINE per far accettare il cambiamento, il video verrà aggiornato; se si desiderano altri cambiamenti si dovrà ripetere la procedura di prima oppure si potrà usare uno dei seguenti comandi:

CA (calcola)

Finora abbiamo solo modificato le caselle; per far adeguare le caselle dipendenti alla modifica, cioè per vedere il cambiamento che ne è seguito, sarà necessario usare questo comando che aggiornerà i dati per poi ritornare alla sezione di input.

SE (settore)

Il tabellone è formato da 9x9 caselle, però è diviso in tre settori da 3x9 caselle (schermate), cioè il settore da 1 a 3, da 4 a 6 e da 7 a 9. Questo comando è importante perché quando si ha bisogno di modificare la casella in un settore diverso da quello in cui stiamo lavorando si dovrà darlo in input.

DX (destra) e SX (sinistra)

Hanno più o meno la stessa funzione di SE però in seguito ad uno di questi non si potrà più fare un input corretto, a meno che non lo si faccia seguire a un comando SE. Servono soprattutto per fare una ricognizione del tabellone di una riga a destra (DX) o a sinistra (SX).

ST (stampa)

Verrà richiesto il nome del lavoro che stiamo eseguendo e il setto-

re da stampare e il suo contenuto sarà trasferito su carta.

SA (salva)

Serve a salvare su cassetta il lavoro finora eseguito: viene richiesto il nome del lavoro e dopo i messaggi per la registrazione verrà eseguito il salvataggio.

CH (change, cambia)

Se si vuole cambiare una casella da commento ad operativa o viceversa si dovrà usare questo comando (attenti a non convertire una casella dipendente).

Ora potete far girare il programma. Se per qualche motivo vi fermate (p.es. per aver premuto il tasto BREAK), niente problemi, battete GOTO 370 e ricominciate.

Il programma gira su tutti gli ZX 81 e ZX 80 nuova ROM con l'espansione da 16 K.

	MARZO 82	APRILE	
ENTRATE	1000000	920000	
TOT. ENTR.	1000000	1500000	
LUCE	50000		
TELEFONO	20000		
CIBO	150000	300000	
VARIE	200000	200000	
USCITE	420000		
RIPORTO	500000		

CARICAMENTO DATI

	MARZO 82	APRILE	
ENTRATE	1000000	920000	
TOT. ENTR.	1000000	1500000	
LUCE	50000		
TELEFONO	20000		
CIBO	150000	300000	
VARIE	200000	200000	
USCITE	420000		
RIPORTO	500000		

COMANDI ? (DX, SX, CA, CH, SE, ST, SA)

	MARZO 82	APRILE	MAGGIO
ENTRATE	1000000	920000	
TOT. ENTR.	1000000	1500000	
LUCE	50000		
TELEFONO	20000		
CIBO	150000	300000	
VARIE	200000	200000	
USCITE	420000		
RIPORTO	500000		

COMANDI ? (DX, SX, CA, CH, SE, ST, SA)

Figura 2. Esempio di esecuzione del programma Minicalc.